Heat transfer - PDSOLVE:

> $restart; with(plots) : with(plottools) :$

layer 1: Ceramic material 1	layer 2: Ceramic material 2	layer 2: Steel shell
λ - thermal conductivity [W/mK] rho - density [kg/m^3] cp - heat capacity [J/kgK]		
> $\lambda 1 := 3.$: $rho1 := 3000$: $cp1 := 1000$: $a1 := \frac{\lambda 1}{rho1 \cdot cp1}$; $a1 := 0.000001000000000$ **(1.1)**	> $\lambda 2 := 1.2$: $rho2 := 2600$: $cp2 := 1100$: $a2 := \frac{\lambda 2}{rho2 \cdot cp2}$; $a2 := 4.195804195\ 10^{-7}$ **(1.2)**	> $\lambda 3 := 44.$: $rho3 := 7850$: $cp3 := 500$: $a3 := \frac{\lambda 3}{rho3 \cdot cp3}$; $a3 := 0.00001121019108$ **(1.3)**
layer 1: thickness = 0,2 m	layer 2: thickness = 0,064 m	layer 2: thickness = 0,03 m

Thermal diffusivity as a piecewise:

> $a := piecewise(x < 0.2, a1, 0.2 \le x < 0.264, a2, a3)$;

$$a := \begin{cases} 0.000001000000000 & x < 0.2 \\ 4.195804195\ 10^{-7} & 0.2 \le x \textbf{ and } x < 0.264 \\ 0.00001121019108 & otherwise \end{cases} \quad \textbf{(1.4)}$$

PDE to solve:

> $PDE := \frac{\partial}{\partial t} T(x,t) = a \left(\frac{\partial^2}{\partial x^2} T(x,t) \right)$;

$$PDE := \frac{\partial}{\partial t} T(x,t) = \left(\begin{cases} 0.000001000000000 & x < 0.2 \\ 4.195804195\ 10^{-7} & 0.2 \le x \textbf{ and } x < 0.264 \\ 0.00001121019108 & otherwise \end{cases} \right) \left(\frac{\partial^2}{\partial x^2} T(x,t) \right) \quad \textbf{(1.5)}$$

Hot face temperature:

> $Thf := 1200$:

Initial and ambient temperature:

> $T0 := 298$:

Overall thickness:

> $L := 0.294$:

Heat convection:

> $h := 10$:

Emmisivity:

> $\epsilon := 0.85$:

Stefan Boltzman - constant

> $\sigma := 5.6697 \cdot 10^{-8}$:

INITIAL AND BOUNDARY CONDITIONS :

> $IBC := \{T(x,0) = T0, T(0,t) = Thf, \lambda 3 \cdot D_1(T)(L,t) = \left(-h \cdot (T(L,t) - T0) + \sigma \cdot \epsilon \cdot \left(T(L,t)^4 - T0^4\right)\right)\}$

$$IBC := \{44.\ D_1(T)(0.294,t) = -10\ T(0.294,t) + 2599.947090 + 4.819245000\ 10^{-8}\ T(0.294,t)^4, T(0,t) = 1200, T(x,0) = 298\} \quad \textbf{(1.6)}$$

PDE:

> $sol := pdsolve(PDE, IBC, numeric, timestep = 50)$:

Plot:

```
> p1 := sol:-plot(t = 72000, thickness = 3, colour = red) :
  p2 := sol:-plot(t = 36000, gridlines = true, axes = boxed, thickness = 3, colour = black) :
  w1 := rectangle([0, 1500], [0.2, 278], thickness = 1, colour = yellow, transparency = 0.9) :
  w2 := rectangle([0.2, 1500], [0.264, 278], thickness = 1, colour = green, transparency = 0.9) :
  w3 := rectangle([0.264, 1500], [0.294, 278], thickness = 1, colour = red, transparency = 0.9) :
  display(p1, p2, w1, w2, w3);
```



> sol:-animate(t = 0 ..72000, frames = 60, thickness = 3, axes = boxed, gridlines = true);



> VAL1 := sol :- value( ); VAL1(0.2, 36000);
VAL2 := sol :- value( ); VAL2(0.264, 36000);
VAL3 := sol :- value( ); VAL3(0.294, 36000);

VAL1 := **proc**( ) ... **end proc**

[x = 0.2, t = 36000., T(x, t) = 754.775086197981]

VAL2 := **proc**( ) ... **end proc**

[x = 0.264, t = 36000., T(x, t) = 706.354883435071]

VAL3 := **proc**( ) ... **end proc**

[x = 0.294, t = 36000., T(x, t) = 702.953275053809] **(1.7)**

FEA System Comsol Multiphisics 4.0a.Model 1D, Number of elements = 128. Transient analysis. Solution for the same geometric, initial and boundary condition.

Solution for time = 3600..72000, step 3600 [second]



Solution for time = 72000 [second]



Line graph: Tem perature (K
1200
1100
1000
900
800
700
600
500
0
0.05
0.1
0.15
0.2
0.25
0.3